
;------------------------------------------------------------------------------
; Program Name : Bridge.logo
; Written By : Herbert Shiu
; Language : LOGO
; Interpreter : UCBLogo
; Output :
;       This program is used to solve the crossing bridge problem.
;       (See Problem Description)
; Problem Description :
;       Four men have to cross a bridge with 17 minutes with following
;       Conditions:
;       a) The bridge can support at most 2 men at a time.
;       b) Crossing a bridge requires a lamp and there is only one lamp.
;       c) The times used to cross the bridge are 1, 2, 5, 10 minutes
;          respectively.
;       d) If two men are crossing the bridge, the time required is the
;          longer one.
; Implications :
;       a) From the above description, the lamp must be passed to and back the
;          bridge.
;       b) Two men cross the bridge from source to target and one men cross
;          the bridge back to the source for bringing the lamp back.
; Usage :
;       Enter "go" at the prompt. And a sample input to the programs is:
;       First : "1 2 5 10" as time list
;       Second : "17" as time limit
; Modification History :
;       09-Jan-1999     Written.
;------------------------------------------------------------------------------
; Copyright 1999 Herbert Shiu. All right reserved.
;------------------------------------------------------------------------------

TO CrossBridge :Source :Target :Limit :Action :Direction
; Output : Prepare data for next step

; If no time left, just quit
        IF :Limit < 0 [STOP]
; If source become empty, i.e., successful, print the solution
        IF :Source = [] ~
        [       PRINT FormatSolution :Action :Limit
                STOP
        ]
; Prepare arguments according to the direction of next step
        IFELSE :Direction = 1 ~
                [CrossTo :Source :Target :Limit :Action GetTwoItems :Source] ~
                [CrossBack :Source :Target :Limit :Action GetItem :Target]
END

TO CrossBack :Source :Target :Limit :Action :BeingTest
; Output : Prepare the environment according to the current going back action.
;          for next steps. If there are other possible actions, try them.

; If no available action, just quit
        IF :BeingTest = [] [STOP]

; Set environment according to the current going back action
        CrossBridge ~
                Union ~
                        :Source ~
                        FIRST :BeingTest ~
                SetDifference ~
                        :Target ~
                        FIRST :BeingTest ~
                :Limit - FIRST FIRST :BeingTest ~
                LPUT ~
                        FIRST :BeingTest ~
                        :Action ~
                1
; Try again with another action
        CrossBack :Source :Target :Limit :Action BUTFIRST :BeingTest
END

TO CrossTo :Source :Target :Limit :Action :BeingTest
; Output : Prepare the environment according to the current going to action.
;          for next steps. If there are other possible actions, try them.

; If no available way, just quit
        IF :BeingTest = [] [STOP]
; Prepare the environment according to the current going to action
        CrossBridge ~
                SetDifference ~
                        :Source ~
                        FIRST :BeingTest ~
                Union ~
                        :Target ~
                        FIRST :BeingTest ~
                :Limit - Max ~
                        FIRST FIRST :BeingTest ~
                        FIRST BUTFIRST FIRST :BeingTest ~
                LPUT ~
                        FIRST :BeingTest ~
                        :Action ~
                -1
; Try again with another action
        CrossTo :Source :Target :Limit :Action BUTFIRST :BeingTest
END

TO FormatSolution :Solution :TimeLeft
; Output : Format the solution to be displayed

; If the end of solution is reached, just return the time left.
        IF :Solution = [] [OUTPUT SENTENCE ": :TimeLeft]
; Format the current action accordingly
        IFELSE (COUNT FIRST :Solution) = 2 ~
                [OUTPUT FPUT ~
                        SENTENCE "TO: FIRST :Solution ~
                        FormatSolution BUTFIRST :Solution :TimeLeft] ~
                [OUTPUT FPUT ~
                        SENTENCE "BACK: FIRST :Solution ~
                        FormatSolution BUTFIRST :Solution :TimeLeft]
END

TO GetItem :InputList

; Output : Generate a list with elements are lists of one element in the
;          input list.
        IF :InputList = [] [OUTPUT []]
        OUTPUT FPUT ~
                FPUT FIRST :InputList [] ~
                GetItem BUTFIRST :InputList
END

To GetTwoItems :InputList
; Output : Generate a list with all possible combinations of two items from
;          the input list.
        IFELSE :InputList = [] ~
                [OUTPUT []] ~
                [OUTPUT ~
                        Union ~
                                GenPairs FIRST :InputList BUTFIRST :InputList ~
                                GetTwoItems BUTFIRST :InputList]
END

TO Go
; Output : Get input from user and prepare arguments for calling the
;          problem solving predicate.
        (LOCAL "TimeList "TimeLimit)
        PRINT [Enter the list of time required:]
        MAKE "TimeList READLIST
        PRINT [Enter the total time limit:]
        MAKE "TimeLimit FIRST READLIST
        CrossBridge :TimeList [] :TimeLimit [] 1
END

TO Max :A :B
; Output : Output the maximum of two items
        IFELSE :A > :B ~
                [OUTPUT :A] ~
                [OUTPUT :B]
END

TO GenPairs :Item :InputList
; Output : Return a list which pair up each item in input list with the
;          input item.
        IFELSE :InputList = [] ~
                [OUTPUT []] ~
                [OUTPUT ~
                        FPUT ~
                                LIST :Item FIRST :InputList ~
                                GenPairs :Item BUTFIRST :InputList]
END

TO SetDifference :A :B
; Output : Return the set difference of A by B
        IF :A = [] ~
                [OUTPUT []]
        IFELSE MEMBERP FIRST :A :B ~
                [OUTPUT SetDifference BUTFIRST :A :B] ~
                [OUTPUT FPUT FIRST :A SetDifference BUTFIRST :A :B]
END

TO Union :A :B

; Output : Return the union of sets A and B
        IF :B = [] ~
                [OUTPUT :A]
        IFELSE MEMBERP LAST :B :A ~
                [OUTPUT Union :A BUTLAST :B] ~
                [OUTPUT LPUT LAST :B Union :A BUTLAST :B]
END

;------------------------------------------------------------------------------
; End of file
;------------------------------------------------------------------------------


